Chapter 1, Polynomial Rings

February 22, 2020

[2]:
```
--initial setup; author: Tyler Zhu
--%mode=pretty
--%timeout=10
```

```
[magic succeeded] mode = pretty
[magic succeeded] timeout = 10
```

1 Section 1.1: Ideals

In Example 1.4, we look at the following two ideals over $\mathbb{Q}[x]$:

$$I = \langle x^3 + 6x^2 + 12x + 8\rangle \text{ and } J = \langle x^2 + x - 2\rangle.$$

Let's try computing their intersection, sum, product, and quotient.

[3]:
```
R = QQ[x] -- set our base ring
```

[7]:
```
I = ideal((x+2)^3)
J = ideal((x+2)*(x-1))
print(concatenate{"I: ", toString I})
print(concatenate{"J: ", toString J})
```

```
I: ideal(x^3+6*x^2+12*x+8)

J: ideal(x^2+x-2)
```

[10]:
```
InJ = intersect(I,J)
print(InJ)
factor InJ_0 -- get the 0th (in this case, only) generator, factored
```

```
        4     3     2
ideal(- x  - 5x  - 6x  + 4x + 8)
```

[12]:
```
IplusJ = I+J
ideal(gens(gb IplusJ))
```

[15]:
```
IprodJ = I*J
print(IprodJ)
factor IprodJ_0
```

```
       5     4      3     2
ideal(x  + 7x  + 16x  + 8x  - 16x - 16)
```

[18]:
```
ImodJ = I:J
print(ImodJ)
factor ImodJ_0
```

```
       2
ideal(x  + 4x + 4)
```

In Example 1.7, we consider the ideal generated by the partial derivatives of the cubic $f = 2xyz - x^2 - y^2 - z^2 + 1$. This is the ideal

$$I = \langle \frac{\partial f}{\partial x}, \frac{\partial f}{\partial y}, \frac{\partial f}{\partial z} \rangle = \langle yz - x, xz - y, xy - z \rangle \subset \mathbb{R}[x, y, z].$$

[19]:
```
R = QQ[x,y,z, MonomialOrder => GLex] --doing over QQ since CAS can't represent
 ↪RR
```

[20]:
```
f = 2*x*y*z-x^2-y^2-z^2+1
```

[21]:
```
vars R
```

[24]:
```
partials = diff(vars R, f)
print(partials)
I = ideal(partials)
```

```
| 2yz-2x 2xz-2y 2xy-2z |
```

[25]:
```
p = primaryDecomposition I
```

[26]:
```
claim = ideal(p_1, p_2, p_3, p_4)
```

[27]:
```
f % claim == 0 -- test membership
```

[28]:
```
ideal(gens gb claim)
```

2 Section 1.2: Grobner Bases

In the beginning of the section, we make the following claim:

$$\langle 2x+3y+5z+7, 11x+13y+17z+19, 23x+29y+31z+37\rangle = \langle 7x{-}16, 7y+12, 7z+9\rangle.$$

[29]:
```
I = ideal(2*x+3*y+5*z+7, 11*x+13*y+17*z+19, 23*x+29*y+31*z+37)
```

[30]:
```
gens gb I
```

Example 1.18 presents us with multiple applications of Grobner Bases. Here, we provide the code for following such applications. Macaulay2 by default employs a graded reverse lexicographic monomial ordering, and our variables are ordered $x \succ y \succ z$ by order of definition.

[31]:
```
R = QQ[x,y,z, MonomialOrder=>Lex]
```

For $n = 1$, computing the reduced Grobner basis means computing the greatest common divisor of the input polynomials:

[32]:
```
F = ideal(x^3-6*x^2-5*x-14, 3*x^3 +8*x^2 +11*x+10, 4*x^4 +4*x^3 +7*x^2-x-2)
```

[33]:
```
G = gens gb F
```

We can look at the ideal $\mathcal{F} = \{xy{-}z, xz{-}y, yz{-}x\}$ and compute its Grobner basis as well as its standard monomials

[36]:
```
F = ideal(x*y-z,x*z-y,y*z-x)
print(F)
G = gens gb F
```

```
ideal (x*y - z, x*z - y, - x + y*z)
```

[37]:
```
staircase = ideal leadTerm F
```

[39]:
```
standardMonomials = R/staircase
basis standardMonomials
```

This input is a curve in the (y, z)-plane parametrized by two cubics in one variable x. We write this as $\mathcal{F} = \{y{-}x^3 + 4x, z{-}x^3{-}x + 1\}$. The Grobner basis has the implicit equation of this curve as its second element.

[41]:
```
use R -- reset the global ring back to R = QQ[x,y,z]
F = ideal(y-x^3+4*x, z-x^3-x+1)
```

[42]:
```
G = gens gb(F)
```

Let z be the sum of $x = \sqrt[3]{7}$ and $y = \sqrt[4]{5}$. We encode this in the set $\mathcal{F} = \{x^3{-}7, y^4{-}5, z{-}x{-}y\}$. The real number $z = \sqrt[3]{7} + \sqrt[4]{5}$ is algebraic of degree 12 over $\mathbb{Q}$. Its minimal polynomial is the first element in the Grobner basis.

[43]:
```
F = ideal(x^3-7, y^4-5, z - x - y)
```

[44]:
```
G = gens gb F
```

The elementary symmetric polynomials $\mathcal{F} = \{x+y+z, xy+xz+yz, xyz\}$ have the reduced Grobner basis $G = \{x+y+z, y^2+yz+z^2, z^3\}$. There are six standard monomials. The quotient $\mathbb{Q}[x,y,z]/I$ is the regular representation of the symmetric group S_3.

[45]:
```
F = ideal(x+y+z,x*y+x*z+y*z, x*y*z)
```

[46]:
```
G = gens gb F
```

[49]:
```
staircase = ideal leadTerm F
standardMonomials = R/staircase
basis standardMonomials
```

[50]:
```
S = R/F
```

3 Section 1.3: Dimension and Degree

(Example 1.32) Let $n = 2m$ be even and consider the monomial ideal $I = \langle x_1x_2, x_3x_4, x_5x_6, \ldots, x_{2m-3}x_{2m-2}, x_{2m-1}x_{2m}\rangle$. The dimension of I equals m and the degree of I equals $2m$. It is instructive to work out the Hilbert series and the Hilbert polynomial of I for $m = 3, 4$.

[51]:
```
R = QQ[x1,x2,x3,x4,x5,x6]
```

[52]:
```
I = ideal(x1*x2, x3*x4, x5*x6)
```

[53]:
```
dim I
```

[54]:
```
degree I
```

[55]:
```
poincare I
```

[56]:
```
numerator reduceHilbert hilbertSeries I
```